

SAS Lab 14 For Statistics 514

Topics:

Chapter 32. Response Surface Methodology

```
*Attendance 14, 32.2, mice roc;
*two-factor rotatable composite design;
DATA mice;
  INPUT roc x1 x2;
  X12 = X1*X2;
  X11 = X1*X1;
  X22 = X2*X2;
DATALINES;
7.6  -1  -1
8.2   1  -1
8.0  -1   1
9.2   1   1
5.4  -1.414 0
5.9   1.414 0
6.3   0  -1.414
8.5   0   1.414
7.3   0   0
7.3   0   0
8.4   0   0
9.5   0   0
;
PROC REG DATA=mice;
  TITLE '32.2 two-factor rotatable composite design, mice';
  MODEL roc = x1 x2 x12 x11 x22;
  OUTPUT OUT=miceout PREDICTED=PRED RESIDUAL=RESID;
RUN;
PROC PRINT DATA=miceout;
  TITLE '32.2, residuals, mice';
  VAR x1 x2 x12 x11 x22 roc PRED RESID;
RUN;
PROC GPLOT DATA=miceout;
  TITLE '32.2, residuals, topping';
  PLOT RESID*PRED;
RUN;
PROC RSREG DATA=mice;
  TITLE '32.2 lack of fit test';
  MODEL roc = x1 x2 x12 x11 x22 / covar=5 lackfit;
RUN;
QUIT;
```

```

*Attendance 14,32.4,mice roc;
*plots/analysis of two-factor rotatable composite design;
DATA mice;
  INPUT roc x1 x2;
        X12 = X1*X2;
        X11 = X1*X1;
        X22 = X2*X2;
DATALINES;
7.6  -1  -1
8.2   1  -1
8.0  -1   1
9.2   1   1
5.4  -1.414 0
5.9   1.414 0
6.3   0  -1.414
8.5   0   1.414
7.3   0   0
7.3   0   0
8.4   0   0
9.5   0   0
;
PROC REG DATA=mice noprint;
  TITLE '32.2 two-factor rotatable composite design, mice';
  MODEL roc = x1 x2 x12 x11 x22;
  OUTPUT OUT=miceout PREDICTED=PRED RESIDUAL=RESID;
RUN;
proc g3grid data=miceout out=surfacegrid;
  grid x2*x1=roc / axis1=-1.41 to 1.41 by .235
        axis2=-1.41 to 1.41 by .235;
run;
proc g3d data=surfacegrid;
  title '32.2 response surface';
  plot x2*x1=roc;
run;
proc g3grid data=miceout out=surfacecontour;
  grid x2*x1=roc / axis1=-1.41 to 1.41 by .235
        axis2=-1.41 to 1.41 by .235
        join;
run;
proc gcontour data=surfacecontour;
  title '32.2 contour of response surface';
  format roc 2.0;
  plot x2*x1=roc;
run;
QUIT;

```

```

*Practice quiz 7, 32.13, whipped topping;
*two-factor rotatable composite design;
DATA topping;
  INPUT percentincrease x1 x2;
  X12 = X1*X2;
  X11 = X1*X1;
  X22 = X2*X2;
DATALINES;
144 -1 -1
192 1 -1
112 -1 1
215 1 1
126 -1.414 0
179 1.414 0
174 0 -1.414
176 0 1.414
196 0 0
191 0 0
177 0 0
195 0 0
;
PROC REG DATA=topping outest=est;
  TITLE '32.13(a) two-factor rotatable composite design, topping';
  MODEL percentincrease = x1 x2 x12 x11 x22;
  OUTPUT OUT=toppingout PREDICTED=PRED RESIDUAL=RESID;
RUN;
PROC PRINT DATA=toppingout;
  TITLE '32.13(b), residuals, topping';
  VAR x1 x2 x12 x11 x22 percentincrease PRED RESID;
RUN;
PROC GPLOT DATA=toppingout;
  TITLE '32.13(b), residuals, topping';
  PLOT RESID*PRED;
RUN;
PROC RSREG DATA=topping;
  TITLE '32.13(c) lack of fit test';
  MODEL percentincrease = x1 x2 x12 x11 x22 / covar=5 lackfit;
RUN;
proc g3grid data=toppingout out=surfacegrid;
  grid x2*x1=percentincrease / axis1=-1.41 to 1.41 by .235
  axis2=-1.41 to 1.41 by .235;
run;
proc g3d data=surfacegrid;
  title '32.13(d) response surface';
  plot x2*x1=percentincrease;
run;
proc g3grid data=toppingout out=surfacecontour;
  grid x2*x1=percentincrease / axis1=-1.41 to 1.41 by .235
  axis2=-1.41 to 1.41 by .235
  join;
run;
proc gcontour data=surfacecontour;
  title '32.13(d) contour of response surface';
  format percentincrease 2.0;
  plot x2*x1=percentincrease;
run;
PROC IML;
  TITLE '32.13(e) CI for mean response';
  USE topping;
  read all var {x1' x2' x12' x11' x22} into x;
  READ ALL VAR {percentincrease} INTO Y;
  N = NROW(x);
  X = J(N,1,1)||x;
  P = NCOL(X);
  B = INV(X'*X)*X'*Y;
  H = X*INV(X'*X)*X;
  SSE = Y'*(I(N) - H)*Y;
  DFE = N - P;
  MSE = SSE/DFE;
  XH = { 1,
  1.22,
  1.16,
  1.4152,
  1.4884,
  1.3456};
  YHAT = XH*B;
  S2B = MSE*INV(X'*X);
  SY = sqrt(XH*S2B*XH);
  PRINT YHAT SY;
RUN;
QUIT;

```